

## ภาคผนวก 2

### ผ.2 Source Code ของ Linterp Function

## ผ.2 Source Code ของ Linterp Function

```
Public Function Linterp(Tbl As Range, ColUsage As Integer, xValue As Double) As Variant
    ' linear interpolator / extrapolator

    Dim nRow As Long
    Dim iLo As Long, iHi As Long

    nRow = Tbl.Rows.Count
    If nRow < 2 Or Tbl.Columns.Count < 2 Then
        Linterp = CVErr(xlErrValue)
        Exit Function '----->
    End If

    If xValue < Tbl(1, 1) Then ' xValue < xmin, extrapolate from first two entries
        iLo = 1
        iHi = 2
    ElseIf xValue > Tbl(nRow, 1) Then ' xValue > xmax, extrapolate from last two entries
        iLo = nRow - 1
        iHi = nRow
    Else
        iLo = Application.Match(xValue, Application.Index(Tbl, 0, 1), 1)
        If Tbl(iLo, 1) = xValue Then ' xValue is exact from table
            Linterp = Tbl(iLo, ColUsage)
            Exit Function '----->
        Else ' xValue is between tabulated values, interpolate
            iHi = iLo + 1
        End If
    End If

    Linterp = Tbl(iLo, ColUsage) + (Tbl(iHi, ColUsage) - Tbl(iLo, ColUsage)) * (xValue - Tbl(iLo, 1)) / (Tbl(iHi, 1) - Tbl(iLo, 1))

End Function
```

ปรับปรุงจาก <http://www.excelforum.com/excel-general/667557-trend-between-two-points.html>